

# XLOOKUP

## آموزش جامع به همراه مثالهای کاربردی

### فهرست مطالب

- ۱..... کار کرد تابع.....
- ۱..... آرگومانهای تابع XLOOKUP اکسل.....
- ۲..... مثال اول - ساده ترین شکل تابع.....
- ۳..... مثال دوم - نمایش چند ستون در نتیجه.....
- ۴..... مثال سوم - نمایش نتیجه با چند شرط.....
- ۵..... مثال چهارم - نمایش پیغام مناسب در صورت پیدا نشدن نتیجه.....
- ۶..... مثال پنجم - جستجو در چند محدوده مختلف.....
- ۷..... مثال ششم - جدیدترین و قدیمی ترین تاریخ با XLOOKUP.....
- ۸..... مثال هفتم - نمایش پاداش فروش.....
- ۱۰..... مثال نهم - استفاده از وایلدکارتهای ( \* ؟ ) جستجو بر اساس بخشی از متن.....
- ۱۱..... نکات مهم تابع XLOOKUP.....
- ۱۱..... ویدیو آموزشی و لینکهای مفید.....

## کار کرد تابع

کار اصلی تابع جستجو در یک محدوده و نمایش مقادیر متناظر با داده جستجو شده .

تابع XLOOKUP در واقع نسخه بسیار پیشرفته و انعطاف پذیر از تابع [VLOOKUP](#) است که محدودیتهای زیادی داشت ، اما این تابع در [آفیس ۲۰۲۱](#) و [آفیس ۳۶۵](#) اضافه شده تا نواقص را مرتفع سازد که به خوبی این کار را انجام داده . این تابع در محدوده افقی و عمودی جستجو می کند و از ویلداکارتها ( \* ؟ ) هم پشتیبانی می کند.

## آرگومانهای تابع XLOOKUP اکسل

**=XLOOKUP(lookup lookup\_array return\_array [not\_found] [match\_mode] [search\_mode] )** ۲۰۲۱

عملکرد	آرگومان
مقداری که باید جستجو شود.	lookup
محدوده و یا آرایه ای که میخواهید مورد جستجو باشد.	lookup_array
مقداری که میخواهید برگردانده شود ، مانند شماره ستون در VLOOKUP	return_array
اگر جستجو نتیجه ای نداشت ، این قسمت اعمال می شود.	[not_found]
عدد ۱- مشخص می کند که ابتدا دقیقا به دنبال مقدار آرگومان بگردد و اگر نبود اولین کوچکترین مقدار را لحاظ کند. عدد ۰ مشخص میکند که فقط دقیقا به دنبال مقدار وارد شده در اول باشد. ( پیش فرض ) عدد ۱ مشخص می کند که ابتدا دقیقا به دنبال مقدار آرگومان بگردد و اگر نبود اولین بزرگترین مقدار را لحاظ کند عدد ۲ مشخص میکند که ویلداکارتها را لحاظ نماید . ( برای درک بهتر حتما ویدیو را ببینید)	[match_mode]
عدد ۱ مشخص می کند که جستجو از بالا به پائین باشد ( پیش فرض ) عدد ۱- مشخص میکند که جستجو از پائین به بالا باشد عدد ۲ جستجو از کوچک به بزرگ عدد ۲- جستجو از بزرگ به کوچک ( برای درک بهتر حتما ویدیو را ببینید)	[search_mode]

## مثال اول - ساده ترین شکل تابع

ساده ترین مثال از این تابع را می توان در یک خط فرمول زیر و تصویر دید که به ساده ترین و سریع ترین شکل ممکن ، کار [VLOOKUP](#) را انجام میدهد و همانطور که در مثال میبیند برای این تابع مهم نیست که محدوده و یا سلولی که میخواهید بعد از جستجو نمایش داده شود ، قبل از محدوده جستجو باشد یا بعد از آن.

```
=XLOOKUP(G2,C2:C23,B2:B23)
```

=XLOOKUP(G2,C2:C23,B2:B23)									
	I	G	F	E	D	C	B	A	
1	نام شرکت	کد کالا		موجودی	شرح کالا	کد کالا	نام شرکت	#	
2	نوریستی ورفاهی آبادگران ایران	۲۶۴		۲	کتری برقی	۲۵۱			
3	لابراتوارداروسازی دکترعبیدی			۲۵	پریز آبی برای اتاق	۲۵۲			
4	سرمایه گذاری البرز(هلدینگ			۱۲	لوله استیل	۲۵۳			
5	آلومراد			۵۴	سماور استیل	۲۵۵			
6	کاشی الوند			۳۵	قاشق و چنگال	۲۵۶			
7	معدنی املاح ایران			۵۲	قاشق و چنگال	۲۵۸			
8	آسان پرداخت پرشین			۱۴	پریز آبی برای اتاق	۲۵۹			
9	سرامیکهایصنعتیاردکان			۵۶	کابل مسی پهن	۲۶۰			
10	آبسال			۲۴	کابل آنتن مرکزی	۲۶۱			

باید جستجو شود

محدوده جستجو

محدوده ای که باید نمایش داده شود

## مثال دوم - نمایش چند ستون در نتیجه

در این مثال فرض میکنیم که جدولی داریم شامل کد پرسنلی و کمپانی و فروش و پاداش و می خواهیم با وارد کردن کد پرسنلی ، بقیه اطلاعات نمایش داده شود ، در تابع Vlookup مجبور بودیم سه فرمول جداگانه بنویسیم ولی اینجا با یک خط فرمول به شکل زیر انجام می شود.

نکته : در آرگومان سوم به جای یک ستون ، ستونهای مد نظر خود را انتخاب می کنیم.

=XLOOKUP(B3,C6:C11,D6:F11)					
F	E ↓	D	C	B	A
					1
					2
					3
					4
					5
					6
					7
					8
					9
					10
					11

  

کد پرسنلی	نام	کمپانی	فروش	پاداش
۱۰۲	حسین	Apple Mini	۲۵۰	۲۲۰۰۰

  

نام	کد پرسنلی	کمپانی	فروش	پاداش
فرهاد	۱۰۰	Google	۵۰	۱۴,۰۰۰
ارسلان	۱۰۱	Sony Album?	۱۰۰	۱۶,۰۰۰
حسین	۱۰۲	Apple Mini	۲۵۰	۲۲,۰۰۰
ارسلان	۱۰۳	Microsoft	۲۰۰	۲۰,۰۰۰
محمد	۵۰۰	Apple Mac?	۱۵۰	۱۸,۰۰۰
عباس	۲۰۰۶	google Sheet	۳۰۰	۲۴,۰۰۰

## مثال سوم - نمایش نتیجه با چند شرط

در این مثال فرض کنیم که اسامی دانش آموزان را با درس مربوطه و نمراتی که اخذ شده را داریم حالا می‌خواهیم با انتخاب نام دانش آموز و انتخاب درس، نمره مورد نظر را نمایش دهد که استفاده از [XLOOKUP با چند شرط](#) را می‌خواهیم بررسی کنیم

```
=XLOOKUP(J2,C3:C10,XLOOKUP(J3,D2:G2,D3:G10))
```

=XLOOKUP(J2,C3:C10,XLOOKUP(J3,D2:G2,D3:G10))

#	نام	ریاضی	علوم	فارسی	اجتماعی
۱	نادر بحری	۱۱	۱۸	۱۴	۱۵
۲	فرهاد بحری	۱۳	۱۵	۹	۱۶
۳	ارسلان	۱۱	۱۶	۱۲	۱۹
۵	امیرعلی	۲۰	۱۶	۸	۱۹
۶	محمد	۱۹	۹	۱۴	
۸	ناصر	۱۹	۱۳	۱۶	۱۰
۹	حسین	۱۷	۱۳	۱۵	۱۶
۱۰	رضا	۸	۲۰	۹	۱۸

=XLOOKUP(I3,D2:G2,XLOOKUP(J2,C3:C10,D3:G10))

البته همانطور که در تصویر می‌بینید، از دو روش برای نمایش نتیجه استفاده شده در واقع تفاوتی با هم ندارند و این تابع به صورت تو در تو استفاده شده و در فقط جای آنها تفاوت کرده است. اگر بخواهیم این کار را ترکیب توابع دیگر انجام دهیم می‌توانیم به سراغ [تابع INDEX](#) و تابع [MATCH](#) برویم.

```
=INDEX(D3:G10,MATCH(J2,C3:C10,0),MATCH(J3,D2:G2,0))
```

## مثال چهارم - نمایش پیغام مناسب در صورت پیدا نشدن نتیجه

فرض می کنیم لیستی از کالاها داریم شامل کد کالا و شرح و تعداد و می خواهیم با وارد کردن کد کالا ، شرح آن نمای داده شود که بسیار ساده است و در اولین مثال این پست آموزش داده شده.

اما اگر کد کالایی که به دنبال آن هستیم در لیست کالاها وجود نداشته باشد ، آیا ارور می دهد ؟ تابع VLOOKUP که در این مواقع ارور #N/A را نمایش می داد.

ولی در این تابع توسط آرگومان چهارم که اختیاری هست می توانیم این موضوع را مدیریت کنیم و دقیقا برای این مسئله در نظر گرفته شده است که به تابع بگوئیم اگر مقدار مورد نظر را پیدا نکردی چه کاری انجام بده.

در تصویر زیر ما به XLOOKUP گفته ایم که اگر پیدا نکردی ، عبارت وجود ندارد را نمایش بده . به همین سادگی 😊

```
=XLOOKUP(H2,C3:C10,D3:D10,"وجود ندارد")
```



The screenshot shows an Excel spreadsheet with the following data:

#	کد کالا	شرح کالا	موجودی اولیه
۱	۱۰۰	کیس گرین	۲۰
۲	۱۰۱	مانیتور ال جی ۱۷ اینچ	۳۲
۳	۱۰۲	کیبورد ۸۷۵۰ فراسو	۲۶
۵	۱۰۳	کیس گرین	۲۶
۶	۱۰۴	اسپیکر جنیوس	۲۶
۸	۱۰۵	کارت صدای کریتو	۲۶
۹	۱۰۶	دی ودی رایتر TEAC	۲۴
۱۰	۱۰۷	کیس گرین	۳۵

The formula bar shows: `=XLOOKUP(H2,C3:C10,D3:D10,"وجود ندارد")`. A green arrow points to cell H2, which contains the value ۲۵۰. A small table to the left of the main table shows the lookup value ۲۵۰ and the result "وجود ندارد".

## مثال پنجم - جستجو در چند محدوده مختلف

فرض کنیم که در دو محدوده مختلف اطلاعاتی با ساختار یکسان ولی داده های متفاوت داریم که شامل کد کالا و شرح کالا هستند.

می خواهیم توسط این تابع ابتدا در محدوده اول به دنبال کد کالا بگردیم و اگر وجود داشت که مقدار متناظر با آن را نمایش دهیم و اگر وجود نداشت محدوده بعدی را جستجو کنیم و در واقع چند تابع را با هم استفاده کنیم.

البته در نظر داشته باشید که این محدوده های متفاوت می توانند در یک شیت و یا در چند شیت و یا حتی در فایل های متفاوت وجود داشته باشند.

```
=XLOOKUP(I2,B3:B7,C3:C7,XLOOKUP(I2,E3:E7,F3:F7))
```

=XLOOKUP(I2,B3:B7,C3:C7,XLOOKUP(I2,E3:E7,F3:F7))

کد کالا	شرح کالا	کد کالا	شرح کالا
۱۰۰	کیس گرین	۱۱۱	کالای ۱
۱۰۱	مانیتور ال جی ۱۷ اینچ	۱۱۲	کالای ۲
۱۰۲	کیبورد ۸۷۵۰ فراسو	۱۱۳	کالای ۳
۱۰۳	کیس گرین	۱۱۴	کالای ۴
۱۰۴	اسپیکر جنیوس	۱۱۵	کالای ۵

ابتدا جدول اول را جستجو می کند اگر پیدا نکرد هبه سراغ جدول بعدی می رود که من تا 10 جدول را تست کرده ام

## مثال ششم - جدیدترین و قدیمی ترین تاریخ با XLOOKUP

فرض کنیم که لیستی از کالاها داریم که دارای یک ستون هستند با نام تاریخ خرید و یا تاریخ ورود به انبار ، خب در این شرایط احتمال تکرار کد کالا در تاریخ های مختلف وجود دارد.

حالا ما می خواهیم تاریخ اولین خرید و تاریخ آخرین خرید یک کالا را نمایش دهیم.

```
=XLOOKUP(H2,B3:B11,E3:E11,0,-1)
```

کد کالا	شرح کالا	موجودی اولیه	تاریخ خرید
۱۰۰	کیس گرین	۲۰	۱۴۰۰/۰۱/۲۰
۱۰۱	مانیتور ال جی ۱۷ اینچ	۳۲	۱۴۰۰/۰۱/۲۱
۱۰۲	کیبورد ۸۷۵۰ فراسو	۲۶	۱۴۰۰/۰۱/۲۲
۱۰۳	کیس گرین	۲۶	۱۴۰۰/۰۱/۲۳
۱۰۶	دی ودی رایتر TEAC	۲۴	۱۴۰۰/۰۱/۲۴
۱۰۱	مانیتور ال جی ۱۷ اینچ	۳۵	۱۴۰۰/۰۱/۲۷
۱۰۸	مادربورد ایسوس	۲۹	۱۴۰۰/۰۱/۲۸
۱۰۹	کیبورد ۸۷۵۰ فراسو	۴۹	۱۴۰۰/۰۱/۲۹
۱۰۱	کیبورد ۸۷۵۰ فراسو	۱۰	۱۴۰۰/۰۲/۰۸

  

کد کالا	تاریخ خرید
۱۰۱	۱۴۰۰/۰۱/۲۱
۱۰۱	۱۴۰۰/۰۲/۰۸

در آرگومان ششم عدد -1 یعنی اینکه در ستون E از پائین به بالا جستجو را انجام بده ، پس در این روش اهمیت دارد که حتما ستون تاریخ از کوچک به بزرگ مرتب سازی شده باشد.

```
=XLOOKUP(H2,B3:B11,E3:E11,0,-1)
```

در این روش آرگومان ششم اهمیت بالایی پیدا میکند چرا که مشخص می کند اگر نتیجه تکراری وجود داشت من از بالا به پائین جستجو کنم و اولین نتیجه که پیدا کردم را نمایش بدم یا از پائین به بالا ؟

پس می بینید که در این روش مرتب سازی از کوچک به بزرگ ستون مد نظر بسیار اهمیت دارد 🙌 .

به یاد داشته باشید که مقدار پیشفرض آرگومان ششم عدد ۱ می باشد و از بالا به پائین است و اگر خالی بگذارید به اولین نتیجه متناظر که برسد جستجو متوقف شده و همان را نمایش می دهد مانند

VLOOKUP .



## مثال هفتم - نمایش پاداش فروش

فرض کنیم که لیستی از پرسنل به همراه مقدار فروش آنها داریم و بر اساس میزان فروش می خواهیم برای آنها پاداش در نظر بگیریم.

برای فروشهای زیر ۲۰ هزار تومان ۵ هزار تومان پاداش و فروش بین ۲۰ هزار تا ۳۰ هزار ، مبلغ ۱۰ هزار تومان پاداش و بالای ۳۰ هزار تومان فروش هم مبلغ ۲۰ هزار تومان پاداش در نظر گرفته شود.

در این حالت چون احتمال دارد میزان فروش دقیقا برابر با میزان فروش در نظر گرفته شده نباشد ، پس باید از تابع XLOOKUP جوری استفاده کنیم که اگر مقدار را پیدا نکرد نزدیک ترین مقدار به آن را نمایش دهد.

این کار را می توانیم با آرگومان پنجم این تابع انجام دهیم که در توضیح آرگومانهای تابع XLOOKUP توضیح عملکرد آن را داده ایم . به تصویر زیر دقت کنید.

```
=XLOOKUP (C3, $F$3:$F$5, $G$3:$G$5, $G$5, 1)
```

=XLOOKUP (C3, \$F\$3:\$F\$5, \$G\$3:\$G\$5, \$G\$5, 1)

نام	فروش	پاداش
نادر بحری	۲۹,۹۰۰	۱۰,۰۰۰
فرهاد بحری	۲۹,۹۰۰	۱۰,۰۰۰
ارسلان	۳۱,۰۰۰	۲۰,۰۰۰
امیرعلی	۲۰,۰۰۰	۵,۰۰۰
محمد	۴۱,۰۰۰	۲۰,۰۰۰
ناصر	۲۴,۰۰۰	۱۰,۰۰۰
حسین	۴۶,۰۰۰	۲۰,۰۰۰

فروش	پاداش
۲۰,۰۰۰	۵,۰۰۰
۳۰,۰۰۰	۱۰,۰۰۰
۴۰,۰۰۰	۲۰,۰۰۰

در آرگومان پنجم عدد 1 نشان دهنده این است که اگر مقدار متناظر با جستجو را پیدا نکردی برو سراغ اولین گزینه بزرگتر از مقدار جستجو ، یعنی اگر قرار بود به دنبال عدد 18000 بگردی چون نیست برو سراغ بعدی که نزدیکترین عدد میشود 20000 هزار و معادل آن را نمایش بده ، آرگومان پنجم به صورت پیش فرض عدد 0 هست که یعنی دقیقا دنبال همان مقدار جستجو بگرد .

## مثال هشتم – بالاترین نمره در درس انتخابی برای کیست؟

فرض کنیم که لیستی از دانش آموزان داریم که برای هر کدام و برای هر درس یک نمره وارد شده و می خواهیم با انتخاب درس توسط تابع XLOOKUP نام شخصی که بالاترین نمره را در آن درس داشته به دست آوریم.

بدیهی است که ابتدا باید با تابع MAX بالاترین نمره را به دست آوریم و مقدار به دست آمده را در آرگومانهای تابع قرار دهیم تا به نتیجه برسیم.

```
=XLOOKUP(MAX(XLOOKUP(I3,$D$2:$G$2,$D$3:$G$10)),XLOOKUP(I3,$D$2:$G$2,$D$3:$G$10),$C$3:$C$10)
```

=XLOOKUP(MAX(XLOOKUP(I3,\$D\$2:\$G\$2,\$D\$3:\$G\$10)),XLOOKUP(I3,\$D\$2:\$G\$2,\$D\$3:\$G\$10),\$C\$3:\$C\$10)

#	نام	ریاضی	علوم	فارسی	اجتماعی
۱	نادر بحری	۹	۱۹	۱۱	۹
۲	فرهاد بحری	۱۹	۱۸	۱۲	۱۹
۳	ارسلان	۱۵	۱۶	۱۷	۱۱
۵	امیرعلی	۲۱	۱۵	۱۰	۱۹
۶	محمد	۲۰	۱۶	۱۷	۱۹
۸	ناصر	۱۵	۱۰	۱۳	۱۰
۹	حسین	۱۳	۹	۱۲	۱۹
۱۰	رضا	۱۰	۱۵	۱۶	۹

بالاترین نمره	نام	درس
۲۱	امیرعلی	ریاضی
۱۹	نادر بحری	علوم
۱۷	ارسلان	فارسی
۱۹	فرهاد بحری	اجتماعی

## مثال نهم - استفاده از وایلدکارتهای ( \* ؟ ) جستجو بر اساس بخشی از متن

تابع XLOOKUP همانطور که در ویدیو هم گفته شد از وایلدکارها پشتیبانی می کند که کاراکتر \* یعنی هر چند کاراکتر و هر کاراکتری و کاراکتر ؟ یعنی یک کاراکتر هر کاراکتری و با در نظر داشتن این اصل ، اگر ابتدا و انتهای متن مورد جستجو کاراکتر \* قرار دهیم به نتیجه خواهیم رسید.

البته شایان ذکر است که برای انجام این مثال تابع [FILTER](#) قطعاً مناسب تر است و همه نتایج شامل را نمایش می دهد و این قسمت بیشتر جنبه آموزشی امکانات این تابع را دارد.

```
=XLOOKUP("*"&E3&"*",C3:C10,C3:C10,,2)
```

`=XLOOKUP("*"&E3&"*",C3:C10,C3:C10,,2)`

نام	#
علی	۱
امیرعلی	۲
نادر بحری	۳
فرهاد بحری	۴
ارسلان	۵
امیرعلی	۶
علی اصغر	۷
محمد ناصر فرهادی	۸
حسین	۹
رضا	۱۰

نام کامل	نام
امیرعلی	علی

کاراکتر \* مشخص میکند که هر کاراکتری به هر تعدادی که بود ، به خاطر داشته باشید که در همین مثال نام کامل دیگری شامل کلمه علی وجود دارد در ردیف 6 ولی تابع XLOOKUP مقدار اول را نمایش می دهد و برای نمایش همه نتایجی که شامل مقدار تایپی باشد از تابع FILTER باید استفاده کنید.

بسیار بعید می دانم با مشاهده کامل این آموزش باز هم سوالی در خصوص تابع XLOOKUP در ذهن شما باقیمانده باشد ولی اگر چیزی بود که من فراموش کرده بودم در این آموزش به اون بپردازم ، حتما در صفحه مربوط به [تابع XLOOKUP در سایت آفیس باز](#) با ما در میان بگذارید.

## نکات مهم تابع XLOOKUP

- تابع [XLOOKUP](#) هم با محدوده عمودی کار می کند و هم با محدوده افقی.
- محدوده جستجو و محدوده ای برگردانده می شود باید همخوان باشند وگرنه ارور #VALUE می دهد
- اگر بین دو فایل جداگانه اکسل از تابع XLOOKUP استفاده کنید باید هر دو فایل باز باشند و در غیر اینصورت ارور #Ref
- این تابع فقط در ورژن ۳۶۵ و یا [توابع آفیس ۲۰۲۱](#) وجود دارد
- اگر بخواهید بر اساس جستجوی شما همه نتایج منطبق را نمایش دهد باید از [تابع FILTER](#) استفاده نمائید

## ویدیو آموزشی و لینکهای مفید

۱. [مشاهده ویدیو آموزش XLOOKUP](#)
۲. [مشاهده لیست توابع و آموزشهای آن](#)
۳. [پروژه های آماده اکسل به همراه آموزش ۰ تا ۱۰۰](#)
۴. [پکیج کامل آموزش اکسل در قالب فلش مموری](#)